

## D02NHF – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

D02NHF is a forward communication routine for integrating stiff systems of implicit ordinary differential equations coupled with algebraic equations when the Jacobian is a banded matrix.

### 2 Specification

```

SUBROUTINE D02NHF(NEQ, NEQMAX, T, TOUT, Y, YDOT, RWORK, RTOL,
1             ATOL, ITOL, INFORM, RESID, YSAVE, NY2DIM, JAC,
2             WKJAC, NWKJAC, JACPVT, NJCPVT, MONITR, LDERIV,
3             ITASK, ITRACE, IFAIL)
  INTEGER      NEQ, NEQMAX, ITOL, INFORM(23), NY2DIM, NWKJAC,
1             JACPVT(NJCPVT), NJCPVT, ITASK, ITRACE, IFAIL
  real        T, TOUT, Y(NEQMAX), YDOT(NEQMAX),
1             RWORK(50+4*NEQMAX), RTOL(*), ATOL(*),
2             YSAVE(NEQMAX,NY2DIM), WKJAC(NWKJAC)
  LOGICAL      LDERIV(2)
  EXTERNAL     RESID, JAC, MONITR

```

### 3 Description

D02NHF is a general purpose routine for integrating the initial value problem for a stiff system of implicit ordinary differential equations coupled with algebraic equations, written in the form,

$$A(t, y)y' = g(t, y).$$

It is designed specifically for the case where the resulting Jacobian is a banded matrix (see description of the argument JAC in Section 5).

Both interval and step oriented modes of operation are available and also modes designed to permit intermediate output within an interval oriented mode.

An outline of a typical calling program for D02NHF is given below. It calls the banded matrix linear algebra setup routine D02NTF, and the Backward Differentiation Formula (BDF) integrator setup routine D02NVF, and its diagnostic counterpart D02NYF.

```

C
C   declarations
C
      EXTERNAL RESID, JAC, MONITR
      .
      .
      .
      IFAIL = 0
      CALL D02NVF(..., IFAIL)
      CALL D02NTF(NEQ, NEQMAX, JCEVAL, ML, MU, NWKJAC, NJCPVT,
+ RWORK, IFAIL)
      IFAIL = -1
      CALL D02NHF(NEQ, NEQMAX, T, TOUT, Y, YDOT, RWORK, RTOL,
+ ATOL, ITOL, INFORM, RESID, YSAVE, NY2DIM, JAC, WKJAC,
+ NWKJAC, JACPVT, NJCPVT, MONITR, LDERIV, ITASK, ITRACE,
+ IFAIL)
      IF (IFAIL.EQ.1 .OR. IFAIL.GE.14) STOP

```

```

IFAIL = 0
CALL D02NYF(...)
      .
      .
      .
STOP
END

```

The linear algebra setup routine D02NTF and one of the integrator setup routines, D02MVF, D02NVF or D02NWF, must be called prior to the call of D02NHF. The integrator diagnostic routine D02NYF may be called after the call to D02NHF. There is also a routine, D02NZF, designed to permit the user to change step size on a continuation call to D02NHF without restarting the integration process.

## 4 References

None.

## 5 Parameters

- 1: NEQ — INTEGER *Input*  
*On entry:* the current trial value of the eigenvalue parameter  $\lambda$ .
- 2: NEQMAX — INTEGER *Input*  
*On entry:* a bound on the maximum number of equations to be solved during the integration.  
*Constraint:* NEQMAX  $\geq$  NEQ.
- 3: T — *real* *Input/Output*  
*On entry:* the value of the independent variable,  $t$ . The input value of T is used only on the first call as the initial point of the integration.  
*On exit:* the value at which the computed solution  $y$  is returned (usually at TOUT).
- 4: TOUT — *real* *Input/Output*  
*On entry:* the next value of  $t$  at which a computed solution is desired. For the initial  $t$ , the input value of TOUT is used to determine the direction of integration. Integration is permitted in either direction (see also ITASK).  
*Constraint:* TOUT  $\neq$  T.  
*On exit:* normally unchanged. However when ITASK = 6, then TOUT contains the value of T at which initial values have been computed without performing any integration. See descriptions of ITASK and LDERIV below.
- 5: Y(NEQMAX) — *real* array *Input/Output*  
*On entry:* the values of the dependent variables (solution). On the first call the first NEQ elements of  $y$  must contain the vector of initial values.  
*On exit:* the computed solution vector, evaluated at  $t$  (usually  $t =$  TOUT).
- 6: YDOT(NEQMAX) — *real* array *Input/Output*  
*On entry:* if LDERIV(1) = .TRUE., YDOT must contain approximations to the time derivatives  $y'$  of the vector  $y$ . If LDERIV(1) = .FALSE., then YDOT need not be set on entry.  
*On exit:* the time derivatives  $y'$  of the vector  $y$  at the last integration point.
- 7: RWORK(50+4\*NEQMAX) — *real* array *Workspace*

**8:** RTOL(\*) — *real* array Input  
**Note:** the dimension of the array RTOL must be at least 1 or NEQ (see ITOL).  
*On entry:* the relative local error tolerance.  
*Constraint:*  $RTOL(i) \geq 0.0$  for all relevant  $i$  (see ITOL).

**9:** ATOL(\*) — *real* array Input  
**Note:** the dimension of the array ATOL must be at least 1 or NEQ (see ITOL).  
*On entry:* the absolute local error tolerance.  
*Constraint:*  $ATOL(i) \geq 0.0$  for all relevant  $i$  (see ITOL).

**10:** ITOL — INTEGER Input  
*On entry:* a value to indicate the form of the local error test. ITOL indicates to D02NHF whether to interpret either or both of RTOL or ATOL as a vector or a scalar. The error test to be satisfied is  $\|e_i/w_i\| < 1.0$ , where  $w_i$  is defined as follows:

ITOL	RTOL	ATOL	$w_i$
1	scalar	scalar	$RTOL(1) \times  y_i  + ATOL(1)$
2	scalar	vector	$RTOL(1) \times  y_i  + ATOL(i)$
3	vector	scalar	$RTOL(i) \times  y_i  + ATOL(1)$
4	vector	vector	$RTOL(i) \times  y_i  + ATOL(i)$

$e_i$  is an estimate of the local error in  $y_i$ , computed internally, and the choice of norm to be used is defined by a previous call to an integrator setup routine.

*Constraint:*  $1 \leq ITOL \leq 4$ .

**11:** INFORM(23) — INTEGER array Workspace  
**12:** RESID — SUBROUTINE, supplied by the user. External Procedure  
 RESID must evaluate the residual

$$r = g(t, y) - A(t, y)y'$$

in one case and

$$r = -A(t, y)y'$$

in another.

Its specification is:

```

SUBROUTINE RESID(NEQ, T, Y, YDOT, R, IRES)
INTEGER          NEQ, IRES
real           T, Y(NEQ), YDOT(NEQ), R(NEQ)
    
```

**1:** NEQ — INTEGER Input  
*On entry:* the number of equations being solved.

**2:** T — *real* Input  
*On entry:* the current value of the independent variable,  $t$ .

**3:** Y(NEQ) — *real* array Input  
*On entry:* the value of  $y_i$ , for  $i = 1, 2, \dots, NEQ$ .

**4:** YDOT(NEQ) — *real* array Input  
*On entry:* the value of  $y'_i$  at  $t$ , for  $i = 1, 2, \dots, NEQ$ .

<p><b>5:</b> R(NEQ) — <i>real</i> array  <i>On exit:</i> R(<i>i</i>) must contain the <i>i</i>th component of <math>r_i</math>, for <math>i = 1, 2, \dots, \text{NEQ}</math>, where,</p> $r = g(t, y) - A(t, y)y' \quad (1)$ <p>or</p> $r = -A(t, y)y' \quad (2)$ <p>and where the definition of <math>r</math> is determined by the input value of IRES.</p> <p><b>6:</b> IRES — INTEGER</p> <p><i>On entry:</i> the form of the residual that must be returned in the array R. If IRES = -1, then the residual defined by equation (2) above must be returned; if IRES = 1, then the residual defined by equation (1) above must be returned.</p> <p><i>On exit:</i> IRES should be unchanged unless one of the following actions is required of the integrator, in which case IRES should be set accordingly.</p> <p>IRES = 2</p> <p>indicates to the integrator that control should be passed back immediately to the calling (sub)program with the error indicator set to IFAIL = 11.</p> <p>IRES = 3</p> <p>indicates to the integrator that an error condition has occurred in the solution vector, its time derivative or in the value of <math>t</math>. The integrator will use a smaller time step to try to avoid this condition. If this is not possible the integrator returns to the calling (sub)program with the error indicator set to IFAIL = 7.</p> <p>IRES = 4</p> <p>indicates to the integrator to stop its current operation and to enter the MONITR routine immediately with parameter IMON = -2.</p>	<p><i>Output</i></p> <p><i>Input/Output</i></p>
------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------	-------------------------------------------------

RESID must be declared as EXTERNAL in the (sub)program from which D02NHF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

- 13:** YSAVE(NEQMAX, NY2DIM) — *real* array *Workspace*
- 14:** NY2DIM — INTEGER *Input*
- On entry:* the second dimension of the array YSAVE as declared in the (sub)program from which D02NHF is called. An appropriate value for NY2DIM is described in the specifications of the integrator setup routines D02MVF, D02NVF and D02NWF. This value must be the same as that supplied to the integrator setup routine.
- 15:** JAC — SUBROUTINE, supplied by the user. *External Procedure*
- JAC must evaluate the Jacobian of the system. If this option is not required, the actual argument for JAC must be the dummy routine D02NHZ. (D02NHZ is included in the NAG Fortran Library and so need not be supplied by the user. Its name may be implementation dependent: see the Users' Note for your implementation for details.) The user indicates whether this option is to be used by setting the parameter JCEVAL appropriately in a call to the linear algebra setup routine D02NTF.

First we must define the system of nonlinear equations which is solved internally by the integrator. The time derivative,  $y'$ , generated internally, has the form

$$y' = (y - z)/(hd)$$

where  $h$  is the current step size and  $d$  is a parameter that depends on the integration method in use. The vector  $y$  is the current solution and the vector  $z$  depends on information from the previous

time steps. This means that  $\frac{d}{dy'}() = \frac{1}{(hd)} \frac{d}{dy}()$ . The system of nonlinear equations that is solved has the form

$$A(t, y)y' - g(t, y) = 0$$

but is solved in the form

$$r(t, y) = 0$$

where  $r$  is the function defined by

$$r(t, y) = (hd)(A(t, y)(y - z)/(hd) - g(t, y)).$$

It is the Jacobian matrix  $\frac{\partial r}{\partial y}$  that the user must supply in the routine JAC as follows:

$$\frac{\partial r_i}{\partial y_j} = a_{ij}(t, y) + (hd) \frac{\partial}{\partial y_j} \left( \sum_{k=1}^{\text{NEQ}} a_{ik}(t, y)y'_k - g_i(t, y) \right)$$

Its specification is:

```

SUBROUTINE JAC(NEQ, T, Y, YDOT, H, D, ML, MU, P)
INTEGER      NEQ, ML, MU
real       T, Y(NEQ), YDOT(NEQ), H, D, P(ML+MU+1,NEQ)

```

- |    |                                                                                                                                  |               |
|----|----------------------------------------------------------------------------------------------------------------------------------|---------------|
| 1: | NEQ — INTEGER                                                                                                                    | <i>Input</i>  |
|    | <i>On entry:</i> the number of equations being solved.                                                                           |               |
| 2: | T — <b>real</b>                                                                                                                  | <i>Input</i>  |
|    | <i>On entry:</i> the current value of the independent variable, $t$ .                                                            |               |
| 3: | Y(NEQ) — <b>real</b> array                                                                                                       | <i>Input</i>  |
|    | <i>On entry:</i> the current solution component $y_i$ , for $i = 1, 2, \dots, \text{NEQ}$ .                                      |               |
| 4: | YDOT(NEQ) — <b>real</b> array                                                                                                    | <i>Input</i>  |
|    | <i>On entry:</i> the derivative of the solution at the current point $t$ .                                                       |               |
| 5: | H — <b>real</b>                                                                                                                  | <i>Input</i>  |
|    | <i>On entry:</i> the current step size.                                                                                          |               |
| 6: | D — <b>real</b>                                                                                                                  | <i>Input</i>  |
|    | <i>On entry:</i> the parameter $d$ which depends on the integration method.                                                      |               |
| 7: | ML — INTEGER                                                                                                                     | <i>Input</i>  |
| 8: | MU — INTEGER                                                                                                                     | <i>Input</i>  |
|    | <i>On entry:</i> the number of sub-diagonals and super-diagonals respectively in the band.                                       |               |
| 9: | P(ML+MU+1,NEQ) — <b>real</b> array                                                                                               | <i>Output</i> |
|    | <i>On exit:</i> elements of the Jacobian matrix $\frac{\partial r}{\partial y}$ stored as specified by the following pseudo code |               |

```

      DO 20 I = 1, NEQ
        J1 = MAX(I-ML, 1)
        J2 = MIN(I+MU, NEQ)
        DO 10 J = J1, J2
          K = MIN(ML+1-I, 0)+J
          P(K, I) = <R/<Y(I, J)
        10 CONTINUE
      20 CONTINUE

```

See also the routine document for F07BDF.

Only non-zero elements of this array need be set, since it is preset to zero before the call to JAC.

JAC must be declared as EXTERNAL in the (sub)program from which D02NHF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

- 16: WKJAC(NWKJAC) — *real* array *Workspace*  
 17: NWKJAC — INTEGER *Input*

*On entry:* the dimension of the array WKJAC as declared in the (sub)program from which D02NHF is called. This value must be the same as that supplied to the linear algebra setup routine D02NTF.

*Constraint:*  $NWKJAC \geq (2m_L + m_U + 1) \times NEQMAX$ , where  $m_L$  and  $m_U$  are the number of sub-diagonals and super-diagonals respectively in the band defined by a call to D02NTF.

- 18: JACPVT(NJCPVT) — INTEGER array *Workspace*  
 19: NJCPVT — INTEGER *Input*

*On entry:* the size of the array JACPVT. This value must be the same as that supplied to the linear algebra setup routine D02NTF.

*Constraint:*  $NJCPVT \geq NEQMAX$ .

- 20: MONITR — SUBROUTINE, supplied by the user. *External Procedure*

MONITR performs tasks requested by the user. If this option is not required, then the actual argument for MONITR must be the dummy routine D02NBY. (D02NBY is included in the NAG Fortran Library and so need not be supplied by the user. Its name may be implementation dependent: see the Users' Note for your implementation for details.)

Its specification is:

```

SUBROUTINE MONITR(NEQ, NEQMAX, T, HLAST, HNEXT, Y, YDOT, YSAVE, R,
1          ACOR, IMON, INLN, HMIN, HMAX, NQU)
INTEGER    NEQ, NEQMAX, IMON, INLN, NQU
  real    T, HLAST, HNEXT, Y(NEQMAX), YDOT(NEQMAX),
1          YSAVE(NEQMAX,*), R(NEQMAX), ACOR(NEQMAX,2),
2          HMIN, HMAX

```

- 1: NEQ — INTEGER *Input*  
*On entry:* the number of equations being solved.
- 2: NEQMAX — INTEGER *Input*  
*On entry:* an upper bound on the number of equations to be solved.
- 3: T — *real* *Input*  
*On entry:* the current value of the independent variable.
- 4: HLAST — *real* *Input*  
*On entry:* the last step size successfully used by the integrator.
- 5: HNEXT — *real* *Input/Output*  
*On entry:* the step size that the integrator proposes to take on the next step.  
*On exit:* the next step size to be used. If this is different from the input value, then IMON must be set to 4.
- 6: Y(NEQMAX) — *real* array *Input/Output*  
*On entry:* the values of the dependent variables,  $y$ , evaluated at  $t$ .  
*On exit:* these values must not be changed unless IMON is set to 2.
- 7: YDOT(NEQMAX) — *real* array *Input*  
*On entry:* the time derivatives  $y'$  of the vector  $y$ .

- 8:** YSAVE(NEQMAX,\*) — *real* array *Input*  
*On entry:* workspace to enable the user to carry out interpolation using either of the routines D02XJF or D02XKF.
- 9:** R(NEQMAX) — *real* array *Input*  
*On entry:* if IMON = 0 and INLN = 3, the first NEQ elements contain the residual vector  $A(t, y)y' - g(t, y)$ .
- 10:** ACOR(NEQMAX,2) — *real* array *Input*  
*On entry:* with IMON = 1, ACOR( $i, 1$ ) contains the weight used for the  $i$ th equation when the norm is evaluated, and ACOR( $i, 2$ ) contains the estimated local error for the  $i$ th equation. The scaled local error at the end of a timestep may be obtained by calling the *real* function D02ZAF as follows
- ```

IFAIL = 1
ERRLOC = D02ZAF(NEQ, ACOR(1,2), ACOR(1,1), IFAIL)
C      CHECK IFAIL BEFORE PROCEEDING

```
- 11:** IMON — INTEGER *Input/Output*  
*On entry:* a flag indicating under what circumstances MONITR was called:
- IMON = -2  
entry from the integrator after IRES = 4 (set in RESID) caused an early termination (this facility could be used to locate discontinuities).
- IMON = -1  
the current step failed repeatedly.
- IMON = 0  
entry after a call to the internal nonlinear equation solver (see below).
- IMON = 1  
the current step was successful.
- On exit:* IMON may be reset to determine subsequent action in D02NHF:
- IMON = -2  
integration is to be halted. A return will be made from the integrator to the calling (sub)program with IFAIL = 12.
- IMON = -1  
allow the integrator to continue with its own internal strategy. The integrator will try up to 3 restarts unless IMON is set  $\neq -1$  on exit.
- IMON = 0  
return to the internal nonlinear equation solver, where the action taken is determined by the value of INLN (see below).
- IMON = 1  
normal exit to the integrator to continue integration.
- IMON = 2  
restart the integration at the current time point. The integrator will restart from order 1 when this option is used. The MONITR provided solution Y will be used for the initial conditions.
- IMON = 3  
try to continue with the same step size and order as was to be used before the call to MONITR. HMIN and HMAX may be altered if desired.
- IMON = 4  
continue the integration but using a new value HNEXT and possibly new values of HMIN and HMAX.

|                                                                                                                                                                                                                                                                                                                                                                                                        |                     |
|--------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|---------------------|
| <p><b>12:</b> INLN — INTEGER</p> <p><i>On exit:</i> the action to be taken by the internal nonlinear equation solver when MONITR is exited with IMON = 0. By setting INLN = 3 and returning to the integrator, the residual vector is evaluated and placed in the array R, and then MONITR is called again. At present this is the only option available: INLN must not be set to any other value.</p> | <i>Output</i>       |
| <p><b>13:</b> HMIN — <i>real</i></p> <p><i>On entry:</i> the minimum step size to be taken on the next step.</p> <p><i>On exit:</i> the minimum step size to be used. If this is different from the input value, then IMON must be set to 3 or 4.</p>                                                                                                                                                  | <i>Input/Output</i> |
| <p><b>14:</b> HMAX — <i>real</i></p> <p><i>On entry:</i> the maximum step size to be taken on the next step.</p> <p><i>On exit:</i> the maximum step size to be used. If this is different from the input value, then IMON must be set to 3 or 4. If HMAX is set to zero, no limit is assumed.</p>                                                                                                     | <i>Input/Output</i> |
| <p><b>15:</b> NQU — INTEGER</p> <p><i>On entry:</i> the order of the integrator used on the last step. This is supplied to enable the user to carry out interpolation using either of the routines D02XJF or D02XKF.</p>                                                                                                                                                                               | <i>Input</i>        |

MONITR must be declared as EXTERNAL in the (sub)program from which D02NHF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

**21:** LDERIV(2) — LOGICAL array *Input/Output*

*On entry:* LDERIV(1) must be set to .TRUE., if the user has supplied both an initial  $y$  and an initial  $y'$ . LDERIV(1) must be set to .FALSE., if only the initial  $y$  has been supplied.

LDERIV(2) must be set to .TRUE. if the integrator is to use a modified Newton method to evaluate the initial  $y$  and  $y'$ . Note that  $y$  and  $y'$ , if supplied, are used as initial estimates. This method involves taking a small step at the start of the integration, and if ITASK = 6 on entry, T and TOUT will be set to the result of taking this small step. LDERIV(2) must be set to .FALSE. if the integrator is to use functional iteration to evaluate the initial  $y$  and  $y'$ , and if this fails a modified Newton method will then be attempted. LDERIV(2) = .TRUE. is recommended if there are implicit equations or the initial  $y$  and  $y'$  are zero.

*On exit:* LDERIV(1) is normally unchanged. However if ITASK = 6 and internal initialisation was successful then LDERIV(1) = .TRUE..

LDERIV(2) = .TRUE., if implicit equations were detected. Otherwise LDERIV(2) = .FALSE..

**22:** ITASK — INTEGER *Input*

*On entry:* the task to be performed by the integrator. The permitted values for ITASK and their meanings are detailed below:

ITASK = 1

normal computation of output values of  $y(t)$  at  $t = TOUT$  (by overshooting and interpolating).

ITASK = 2

take one step only and return.

ITASK = 3

stop at the first internal integration point at or beyond  $t = TOUT$  and return.

ITASK = 4

normal computation of output values of  $y(t)$  at  $t = TOUT$  but without overshooting  $t = TCRIT$ . TCRIT must be specified as an option in one of the integrator setup routines prior to



the first call to the integrator, or specified in the optional input routine prior to a continuation call. TCRIT may be equal to or beyond TOUT, but not before it, in the direction of integration.

ITASK = 5

take one step only and return, without passing TCRIT. TCRIT must be specified as under ITASK = 4.

ITASK = 6

the integrator will solve for the initial values of  $y$  and  $y'$  only and then return to the calling (sub)program without doing the integration. This option can be used to check the initial values of  $y$  and  $y'$ . Functional iteration or a ‘small’ backward Euler method used in conjunction with a damped Newton iteration is used to calculate these values (see LDERIV above). Note that if a backward Euler step is used then the value of  $t$  will have been advanced a short distance from the initial point.

**Note.** If D02NHF is recalled with a different value of ITASK (and TOUT altered), then the initialisation procedure is repeated, possibly leading to different initial conditions.

*Constraint:*  $1 \leq \text{ITASK} \leq 6$ .

### 23: ITRACE — INTEGER

*Input*

*On entry:* the level of output that is printed by the integrator. ITRACE may take the value  $-1$ ,  $0$ ,  $1$ ,  $2$  or  $3$ . If  $\text{ITRACE} < -1$ , then  $-1$  is assumed and similarly if  $\text{ITRACE} > 3$ , then  $3$  is assumed. If  $\text{ITRACE} = -1$ , no output is generated. If  $\text{ITRACE} = 0$ , only warning messages are printed on the current error message unit (see X04AAF). If  $\text{ITRACE} > 0$  then warning messages are printed as above, and on the current advisory message unit (see X04ABF) output is generated which details Jacobian entries, the nonlinear iteration and the time integration. The advisory messages are given in greater detail the larger the value of ITRACE.

### 24: IFAIL — INTEGER

*Input/Output*

*On entry:* IFAIL must be set to  $0$ ,  $-1$  or  $1$ . Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

*On exit:* IFAIL =  $0$  unless the routine detects an error or gives a warning (see Section 6).

**For this routine**, because the values of output parameters may be useful even if IFAIL  $\neq 0$  on exit, users are recommended to set IFAIL to  $-1$  before entry. **It is then essential to test the value of IFAIL on exit.** To suppress the output of an error message when soft failure occurs, set IFAIL to  $1$ .

## 6 Error Indicators and Warnings

Errors or warnings specified by the routine:

IFAIL = 1

An illegal input was detected on entry, or after an internal call to MONITR. If ITRACE  $> -1$ , then the form of the error will be detailed on the current error message unit (see X04AAF).

IFAIL = 2

The maximum number of steps specified has been taken (see the description of optional inputs in the integrator setup routines and the optional input continuation D02NZF).

IFAIL = 3

With the given values of RTOL and ATOL no further progress can be made across the integration range from the current point T. The components  $Y(1), Y(2), \dots, Y(\text{NEQ})$  contain the computed values of the solution at the current point T.

IFAIL = 4

There were repeated error test failures on an attempted step, before completing the requested task, but the integration was successful as far as T. The problem may have a singularity, or the local error requirements may be inappropriate.

IFAIL = 5

There were repeated convergence test failures on an attempted step, before completing the requested task, but the integration was successful as far as T. This may be caused by an inaccurate Jacobian matrix or one which is incorrectly computed.

IFAIL = 6

Some error weight  $w_i$  became zero during the integration (see description of ITOL). Pure relative error control ( $ATOL(i) = 0.0$ ) was requested on a variable (the  $i$ th) which has now vanished. The integration was successful as far as T.

IFAIL = 7

The user-supplied subroutine RESID set its error flag (IRES = 3) continually despite repeated attempts by the integrator to avoid this.

IFAIL = 8

LDERIV(1) = .FALSE. on entry but the internal initialisation routine was unable to initialise  $y'$  (more detailed information may be directed to the current error message unit, see X04AAF).

IFAIL = 9

A singular Jacobian  $\frac{\partial r}{\partial y}$  has been encountered. The user should check his problem formulation and Jacobian calculation.

IFAIL = 10

An error occurred during Jacobian formulation or back-substitution (a more detailed error description may be directed to the current error message unit, see X04AAF).

IFAIL = 11

The user-supplied subroutine RESID signalled the integrator to halt the integration and return (IRES = 2). Integration was successful as far as T.

IFAIL = 12

The user-supplied subroutine MONITR set IMON = -2 and so forced a return but the integration was successful as far as T.

IFAIL = 13

The requested task has been completed, but it is estimated that a small change in RTOL and ATOL is unlikely to produce any change in the computed solution. (Only applies when the user is not operating in one step mode, that is when ITASK  $\neq$  2 or 5).

IFAIL = 14

The values of RTOL and ATOL are so small that the routine is unable to start the integration.

IFAIL = 15

The linear algebra setup routine D02NTF was not called before the call to D02NHF.

## 7 Accuracy

The accuracy of the numerical solution may be controlled by a careful choice of the parameters RTOL and ATOL, and to a much lesser extent by the choice of norm. Users are advised to use scalar error control unless the components of the solution are expected to be poorly scaled. For the type of decaying solution typical of many stiff problems, relative error control with a small absolute error threshold will be most appropriate (that is the user is advised to choose ITOL = 1 with ATOL(1) small but positive).

## 8 Further Comments

The cost of computing a solution depends critically on the size of the differential system and to a lesser extent on the degree of stiffness of the problem. For D02NHF the cost is proportional to  $\text{NEQ} \times (\text{ML} + \text{MU} + 1)^2$  though for problems which are only mildly nonlinear the cost may be dominated by factors proportional to  $\text{NEQ} \times (\text{ML} + \text{MU} + 1)$  except for very large problems.

In general the user is advised to choose the BDF option (setup routine D02NVF) but if efficiency is of great importance and especially if it is suspected that  $\frac{\partial}{\partial y}(A^{-1}g)$  has complex eigenvalues near the imaginary axis for some part of the integration, the user should try the BLEND option (setup routine D02NWF).

## 9 Example

We solve the well-known stiff Robertson problem written as an implicit differential system and in implicit form

$$\begin{aligned} r_1 &= a' + b' + c' \\ r_2 &= 0.04a - 1.0E4bc - 3.0E7b^2 - b' \\ r_3 &= 3.0E7b^2 - c' \end{aligned}$$

exploiting the fact that we can show that  $(a + b + c)' = 0$  for all time. Integration is over the range  $[0, 10.0]$  with initial conditions  $a = 1.0$  and  $b = c = 0.0$  using scalar relative error control and vector absolute error control ( $\text{ITOL} = 2$ ). We integrate using a BDF method (setup routine D02NVF) and a modified Newton method. The Jacobian is calculated numerically and we employ a default monitor, dummy routine D02NBY. We perform a normal integration ( $\text{ITASK} = 1$ ) to obtain the value at  $\text{TOUT} = 10.0$  by integrating past  $\text{TOUT}$  and interpolating. We also illustrate the use of  $\text{ITASK} = 6$  to calculate initial values of  $y$  and  $y'$  and then return without integrating further.

### 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      D02NHF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER       (NOUT=6)
      INTEGER          NEQ, NEQMAX, NRW, NINF, ML, MU, NJCPVT, NWKJAC,
+                    MAXORD, NY2DIM, MAXSTP, MXHNIL
      PARAMETER       (NEQ=3, NEQMAX=NEQ, NRW=50+4*NEQMAX, NINF=23, ML=1,
+                    MU=2, NJCPVT=NEQMAX, NWKJAC=NEQMAX*(2*ML+MU+1),
+                    MAXORD=5, NY2DIM=MAXORD+1, MAXSTP=200, MXHNIL=5)
      real            HO, HMAX, HMIN, TCRT
      PARAMETER       (HO=1.0e-4, HMAX=10.0e0, HMIN=1.0e-10, TCRT=0.0e0)
      LOGICAL         PETZLD
      PARAMETER       (PETZLD=.FALSE.)
*      .. Local Scalars ..
      real            H, HU, T, TCUR, TOLSF, TOUT
      INTEGER          I, IFAIL, IMXER, ITASK, ITOL, ITRACE, NITER, NJE,
+                    NQ, NQU, NRE, NST
*      .. Local Arrays ..
      real            ATOL(NEQMAX), CONST(6), RTOL(NEQMAX), RWORK(NRW),
+                    WKJAC(NWKJAC), Y(NEQMAX), YDOT(NEQMAX),
+                    YSAVE(NEQMAX, NY2DIM)
      INTEGER          INFORM(NINF), JACPVT(NJCPVT)
      LOGICAL         ALGEQU(NEQMAX), LDERIV(2)
*      .. External Subroutines ..
      EXTERNAL        D02NBY, D02NHF, D02NHZ, D02NTF, D02NVF, D02NYF,

```

```

+           RESID, X04ABF
*   .. Executable Statements ..
WRITE (NOUT,*) 'D02NHF Example Program Results'
CALL X04ABF(1,NOUT)
*
*   Set ITASK=6 to provide initial estimates of solution and its
*   time derivative. Default values for the array CONST are used.
*   Use the B.D.F. formulae with a Newton method.
*   Employ scalar relative tolerance and vector absolute tolerance.
*   The Jacobian is evaluated internally.
*   MONITR subroutine replaced by NAG dummy routine D02NBY.
*
T = 0.0e0
TOUT = 10.0e0
ITASK = 6
Y(1) = 1.0e0
Y(2) = 0.0e0
Y(3) = 0.0e0
LDERIV(1) = .FALSE.
LDERIV(2) = .FALSE.
ITOL = 2
RTOL(1) = 1.0e-4
ATOL(1) = 1.0e-6
ATOL(2) = 1.0e-7
ATOL(3) = 1.0e-6
DO 20 I = 1, 6
    CONST(I) = 0.0e0
20 CONTINUE
IFAIL = 0
*
CALL D02NVF(NEQMAX,NY2DIM,MAXORD,'Newton',PETZLD,CONST,TCRIT,HMIN,
+          HMAX,HO,MAXSTP,MXHNIL,'Average-L2',RWORK,IFAIL)
*
CALL D02NTF(NEQ,NEQMAX,'Numerical',ML,MU,NWKJAC,NJCPVT,RWORK,
+          IFAIL)
*
Soft fail and error messages only
ITRACE = 0
IFAIL = 1
*
CALL D02NHF(NEQ,NEQMAX,T,TOUT,Y,YDOT,RWORK,RTOL,ATOL,ITOL,INFORM,
+          RESID,YSAVE,NY2DIM,D02NHZ,WKJAC,NWKJAC,JACPVT,NJCPVT,
+          D02NBY,LDERIV,ITASK,ITRACE,IFAIL)
*
WRITE (NOUT,*)
IF (IFAIL.EQ.0) THEN
    WRITE (NOUT,99999) ' Initial   Y : ', (Y(I),I=1,NEQ)
    WRITE (NOUT,99999) ' Initial YDOT : ', (YDOT(I),I=1,NEQ)
ELSE
    WRITE (NOUT,99997) 'Exit D02NHF with IFAIL = ', IFAIL,
+          ' and T = ', T
    STOP
END IF
*
Use these initial estimates and integrate to TOUT (overshoot and
*
interpolate)
ITASK = 1
IFAIL = 1
TOUT = 10.0e0

```

```

*
  CALL D02NHF(NEQ,NEQMAX,T,TOUT,Y,YDOT,RWORK,RTOL,ATOL,ITOL,INFORM,
+           RESID,YSAVE,NY2DIM,DO2NHZ,WKJAC,NWKJAC,JACPVT,NJCPVT,
+           DO2NBY,LDERIV,ITASK,ITRACE,IFAIL)
*
  IF (IFAIL.EQ.0) THEN
    WRITE (NOUT,*)
    WRITE (NOUT,*)
+   ' X          Y(1)          Y(2)          Y(3)'
    WRITE (NOUT,99998) TOUT, (Y(I),I=1,NEQ)
*
    CALL D02NYF(NEQ,NEQMAX,HU,H,TCUR,TOLSF,RWORK,NST,NRE,NJE,NQU,
+           NQ,NITER,IMXER,ALGEQU,INFORM,IFAIL)
*
    WRITE (NOUT,*)
    WRITE (NOUT,99996) ' HUSED = ', HU, ' HNEXT = ', H,
+   ' TCUR = ', TCUR
    WRITE (NOUT,99995) ' NST = ', NST, ' NRE = ', NRE,
+   ' NJE = ', NJE
    WRITE (NOUT,99995) ' NQU = ', NQU, ' NQ = ', NQ,
+   ' NITER = ', NITER
    WRITE (NOUT,99994) ' Max err comp = ', IMXER
  ELSE
    WRITE (NOUT,*)
    WRITE (NOUT,99997) 'Exit D02NHF with IFAIL = ', IFAIL,
+   ' and T = ', T
  END IF
  STOP
*
99999 FORMAT (1X,A,3(F11.4,2X))
99998 FORMAT (1X,F8.3,3(F13.5,2X))
99997 FORMAT (1X,A,I2,A,e12.5)
99996 FORMAT (1X,A,e12.5,A,e12.5,A,e12.5)
99995 FORMAT (1X,A,I6,A,I6,A,I6)
99994 FORMAT (1X,A,I4)
  END
*
SUBROUTINE RESID(NEQ,T,Y,YDOT,R,IRES)
*
  .. Scalar Arguments ..
  real          T
  INTEGER        IRES, NEQ
*
  .. Array Arguments ..
  real          R(NEQ), Y(NEQ), YDOT(NEQ)
*
  .. Executable Statements ..
  R(1) = -YDOT(1) - YDOT(2) - YDOT(3)
  R(2) = -YDOT(2)
  R(3) = -YDOT(3)
  IF (IRES.EQ.1) THEN
    R(1) = 0.0e0 + R(1)
    R(2) = 0.04e0*Y(1) - 1.0e4*Y(2)*Y(3) - 3.0e7*Y(2)*Y(2) + R(2)
    R(3) = 3.0e7*Y(2)*Y(2) + R(3)
  END IF
  RETURN
  END

```

## 9.2 Program Data

None.

### 9.3 Program Results

D02NHF Example Program Results

WARNING... EQUATION(=I1) AND POSSIBLY OTHER EQUATIONS ARE  
IMPLICIT AND IN CALCULATING THE INITIAL VALUES THE EQNS  
WILL BE TREATED AS IMPLICIT.

IN ABOVE MESSAGE I1 = 1

|                |         |        |        |
|----------------|---------|--------|--------|
| Initial Y :    | 1.0000  | 0.0000 | 0.0000 |
| Initial YDOT : | -0.0400 | 0.0400 | 0.0000 |

WARNING... EQUATION(=I1) AND POSSIBLY OTHER EQUATIONS ARE  
IMPLICIT AND IN CALCULATING THE INITIAL VALUES THE EQNS  
WILL BE TREATED AS IMPLICIT.

IN ABOVE MESSAGE I1 = 1

| X      | Y(1)    | Y(2)    | Y(3)    |
|--------|---------|---------|---------|
| 10.000 | 0.84135 | 0.00002 | 0.15863 |

HUSED = 0.91752D+00 HNEXT = 0.91752D+00 TCUR = 0.10885D+02

NST = 51 NRE = 118 NJE = 14

NQU = 4 NQ = 4 NITER = 68

Max err comp = 3

---